TREE-RING SKELETON PLOTTING BY COMPUTER

JOHN PHILIP CROPPER

Laboratory of Tree-Ring Research
The University of Arizona

ABSTRACT

Skeleton plotting is an established manual technique for representing the relative narrowness of tree rings in a single radius. These plots can be used as a visual aid to crossdating. This paper describes a method for deriving these plots by computer. The method uses a low-pass digital filter, running means, and standard deviations of ring-width measurements.

When the manual and computer plots are compared for the same series, approximately 85% agreement is found.

Examples of results are presented for specimens from sensitive, moderate, and complacent sites. FORTRAN program listings are included for two subroutines for (a) identifying small rings and (b) producing the plot.

La technique du "Skeleton plot" est une méthode manuelle pratiquée pour mettre en évidence la minceur relative des cernes observés suivant un rayon. Les graphiques ainsi obtenus constituent une aide visuelle pour l'interdatation des échantillons. Le présent article décrit une méthode permettant de construire, par ordinateur, ces graphiques. La méthode utilise un filtre digital passe-bas, des moyennes courantes et les déviations standard des mesures de cernes. Lorsque des séries manuelles sont comparées à fournies par l'ordinateur, on constate approximativement 85% de concordance. L'édition des programmes FORTRAN est jointe pour deux sous-routines: (a) identification des petits cernes (b) production du graphique.

Die Anfertigung von sog. "skeleton plots" oder Minimum-Diagrammen ist ein bewährtes manuelles Verfahren zur Veranschaulichung der relativen Breite von schmalen Jahresringen eines Radius. Diese Diagramme können als visuelle Hilfe bei der Synchronisierung von Jahrringfolgen dienen. Der vorliegende Beitrag beschreibt die Möglichkeit, derartige Diagramme mit Hilfe eines Computers anzufertigen. Die Methode benutzt einen Zahlenfilter,, der die kurzwellige und hochfrequente Streuung einer Jahrringfolge eliminiert, ferner die gleitende Mittelwertbildung und die Standardabweichungen der Jahrringbreiten.

Beim Vergleich von "skeleton plots" derselben Jahrringfolgen, die sowohl manuell als auch vom erstellt Computer wurden, ergab sich eine Übereinstimmung von 85%. Es werden Beispiele für Holzproben gezeigt, die von Standorten stammen, wo die Bäume Jahrringfolgen mit starken, mittleren und geringen jährlichen Schwankungen ausbilden. Zudem sind die FORTRAN-Versionen für zwei Unterprogramme enthalten, die schmale Jahrringe erkennen und das Minimum-Diagramm anfertigen.

INTRODUCTION

Over the past ten years the computer and other machines have increased in their importance to the science of dendrochronology. The use of such technical aids has resulted in an ability to process many more specimens with limited manpower. The workers at the University of Washington, College of Forest Resources, have made advances in linking a cassette recorder to a measuring machine and bypassing the necessity (and possibly error-prone step) of card-punching (Brubaker 1979, personal communication). Other workers using X-ray densitometry work with a totally different aspect of the ring (i.e., density rather than total ring width) (Parker et. al. 1976; Polge 1966; Milsom and Hughes 1977). In Virginia at the U. S. Geological Survey, the actual viewing of the wood has been made easier by magnifying the microscope image onto a color television using a close-circuit camera installed into the microscope (Phipps 1979, personal communication). Most of the advancement in the science has been made possible by the speed at which the computer can accomplish complex mathematical manipulations of vast amounts of data. Examples of such programs are the standardizing program INDXA (Fritts et al. 1969) and the response function program (Fritts et al. 1971). Programs that are useful at earlier stages in chronology building have been written and these include computer crossdating programs. Crossdating programs have been in use since 1965 with varying degrees of success. All of them basically use a system involving cross-correlation techniques (Fritts 1963; Parker 1967; Baillie and Pilcher 1973; Wendland 1975).

At the Laboratory of Tree-Ring Research another crossdating aid called skeleton plotting is used. In chronology development at this Laboratory, the recommended procedural steps are to (1) skeleton plot the samples and then (2) go through the pattern-matching process to actually crossdate the wood. Normally only after each ring has been assigned to the calendar year is (3) the wood measured. The author envisages that by changing the order of the above procedure (to 3, 1, 2) and by using a computer program as described below, consistent skeleton plots can be produced with considerable timesaving. This program is *not* for crossdating and does *not* replace the practical experience of many years of examining diverse wood samples, but should be considered an aid to crossdating.

METHOD

Skeleton plotting has been practiced manually at the Laboratory of Tree-Ring Research, Tucson, Arizona, since 1927 (Douglass 1935). This paper demonstrates that in certain circumstances a statistically developed computer skeleton plot can be produced from undated ring widths as an aid to the crossdating procedure. In this procedure, each ring in the tree-ring series is assessed for its narrowness when compared to its immediate neighbors. Each narrow ring is then represented on a linear plot as a line of increasing length proportional to its narrowness. Thus, a series of ring widths can be represented by a piece of graph paper with the significant (narrow) rings indicated. Plotting a series of such lines for different specimens makes matching the significant patterns and thus crossdating the samples considerbly easier (Schulman 1942; Giddings 1942; Schulman 1944; Lyon 1953; Stokes and Smiley 1968; Ferguson 1970).

The author accepts that in some instances the computer skeleton plotting procedure would be of little or no use. For example, when skeleton plotting is used for dating archeological specimens, such as roof timbers or charcoal, then the date itself is the only information the technician is seeking. In such cases the ring widths would not necessarily be measured and a computer-processed skeleton plot would only involve unnecessary work. However, many other dendrochronological analyses, such as dendroclimatology, require measurement of the ring widths in order to form the final chronology.

The skeleton plotting program is designed to accomplish what a dendrochronologist does when he microscopically examines a piece of wood and makes a decision on narrowness based on the comparison of each ring with its immediate neighbors. In programming the computer for this comparison, the running mean and standard deviations are determined, and these two parameters are used to make the decision on relative narrowness. The number of rings averaged and the best critical level for narrowness (the number of standard deviations smaller than a ring must be to be considered narrow) depend to some extent upon the sensitivity (year-to-year variability in ring width) of the specimen.

There is an "end" effect such that narrowness cannot be determined for the rings at the extreme ends of the series. The number of rings involved is equal to the number of years used in the mean minus one; thus, using a mean of five years there would be the loss of two years data from each end. The actual process of computing the standard deviation of a mean of three numbers prevents two small rings occurring consecutively from being recognized as narrow. Thus, two skeleton marks will not appear next to each other on plot (c) (Figures 1, 2, and 3). Because of the potential problems of double small rings occurring together, it is suggested that means of five or more



Figure 1. Series 478041. A moderately sensitive series of white spruce (*Picea Glauca*) from Labrador. (a) graphic plot of ring widths in 100ths of millimeters. (b) manually derived control skeleton plot. (c) - (g) computer-derived skeleton plots using varying numbers of years in the running mean ($\overline{X}$) and varying multiples of standard deviations (S. D.) for the critical levels beyond which a ring is defined as large or small.

years be used for developing the plots. In order to obtain more skeleton marks for use in the dating process, the reduction of the critical level to 0.75 standard deviations considerably increases the number of "significantly small" rings. Even more marks are obtained with a reduction of the critical level to 0.5 standard deviations.

For example, suppose the mean of five rings is 2.0 mm with a standard deviation of 0.75 mm and the critical level is chosen to be 0.5 standard deviations. Then if the central ring is smaller than (2.0 - (0.5 * 0.75)), it is plotted as a narrow ring. If the central ring is greater than (2.0 + (0.5 * 0.75)), it is marked as a large ring (L). If the ring width falls between these two limits, it is ignored and plotted as zero. For small



Figure 2. Series 479122. A highly sensitive series of white fir (*Abies concolor*) from Nevada, U.S.A. (a) - (g) same as Figure 1.

rings, then, the difference between the lower bound of the critical level and the actual ring size is stored in an array. Once the whole series has been analyzed, the smallest ring (i.e., that ring which deviates furthest from the critical level) is scaled such that is will be plotted as a 2 cm line. All the other small rings are subsequently scaled by the same factor before being plotted.



Figure 3. Series 749032. A complacent (nonsensitive) series of Western larch (*Larix occidentalis*) from British Columbia, Canada. (a) - (g) same as Figure 1.

RESULTS

Figure 1 is an example of a core from Labrador, Canada (ID = 478041). Dating the site was straightforward since there were no missing or double rings, and the mean sensitivity of the core was .224 (Duvick 1978). The top plot (a) is a graphic representation of the ring widths. Below that (b) is the skeleton plot derived manually by T. P. Harlan. Mr. Harlan is a highly experienced dendrochronologist at the Laboratory of Tree-Ring Research who specializes in crossdating tree-ring specimens. His manual skeleton plots are used as the control against which the computerized plots should be compared. The next five skeleton plots (c to g) were derived by the objective method described above, and differences between them are due to varying the number of rings considered in the running mean and varying the critical level. Plots (c), (d), and (e) all use a critical level of running mean width minus one standard deviation. The number of rings considered in the running mean is changed from three in (c) to seven in (e). In the final two plots (f and g) the critical level is varied to 0.75 and 0.5 standard deviations from the running mean of five rings in both cases. Using plots (a) and (b) as controls, definite differences can be seen between plots (c), (d), and (e). Basically the same rings are selected as being significantly smaller (or larger "L"); however, there are differences. Plot (g) would be chosen as the most similar to that developed manually: for the 107-year period it contains similar numbers of marks (36 compared to (b)'s 30 marks) of which 21 are in agreement (77.6% agreement).

Figures 2 and 3 show the results of using the same program on a sensitive series (ID = 479122) from Nevada, U.S.A., and a complacent series (ID = 749032) from British Columbia, Canada. Using the combination of a running mean of five years and a critical level of 0.5 standard deviations from the mean, a high degree of agreement with the manually produced control was obtained in these two examples: 85% in the sensitive site (Figure 2) and 81% in the complacent site (Figure 3).

However, on the complacent site (Figure 3) with a high degree of low-frequency trend, the small rings within a period of slow growth are not plotted as being as important as those small rings in fast-growth portions of the core. The small rings in these portions of the data are still correctly identified, but the scaling scheme was biased against them. The scaling scheme uses the difference between the critical level and the actual ring size, and in a period of slow growth this difference can never be very large. In a fast-growth period the potential for larger differences exists. The method used to correct for this bias converts the widths to indices before performing the skeleton plotting. By using a 13-weight low-pass filter (LaMarche and Fritts 1972; Julian and Fritts 1968) on the data a smoothed curve is fit. An index series is obtained by dividing the original data by the resultant low-pass filter values. When skeleton plots of index data are computed, the sensitive series (479122) gives identical results to those from the unaltered series. The less sensitive index series (478041) identifies the same rings as being small, but gives very slightly different weightings to them. The complacent series (479032) also identifies essentially the same rings, but the weightings given on the plot are far superior on the indexed series. Thus, the filtering and indexing procedures have little effect on sensitive series, but are beneficial on more difficult complacent series. It should be noted that the plots in Figure 1, 2, and 3 were all produced using low-pass filter indexed data.

CONCLUSIONS

The most consistent and generally applicable scheme for performing skeleton plotting by computer was found to be: (1) indexing each series by dividing the data by a low-pass filtered curve fit to the same data; and (2) identifying small rings using a running mean of five years and a critical level of 0.5 standard deviations from the mean. The production of skeleton plots is only one step in the process of chronology development. Once the plots have been produced, the patterns of each specimen must be matched (crossdated) and calendar years assigned to them. A relatively simple computer program can be written to take the original tree-ring width data cards and repunch them into a standard format with years assigned. The series should then be standardized and averaged as described by Fritts (1976), possibly using similar programs as described by Graybill (this issue).

For those interested in experimenting with this aid, listings (Figures 4 and 5) of two subroutines are included. Both subroutines are written in ANSI standard FORTRAN to make them machine-independent and to minimize any problems of implementation. The first subroutine (SKEL1) calculates the running means and running standard deviations and identifies the "small" rings. The second subroutine (SKELET) produces the plots utilizing the output from SKEL1, and CALCOMP plotting subroutines that may not be available on all machines.

The input needed for SKEL1 is: (1) an array containing the widths; (2) the number of years to be used in the calculation of the mean; (3) the critical value expressed in fractions of standard deviations; (4) the number of years in the input vector; and (5) in which row of the output matrix to place the results.

The input needed for SKELET is: (1) the output matrix from SKEL1; (2) an array containing the number of data values in each row of the above matrix; (3) an integer variable specifying the number of rows of the input matrix to be plotted (between 1 and 4); (4) an array containing the ID's of each of the data rows being plotted; and (5) a general heading to be placed on the top of the plot.

ACKNOWLEDGEMENTS

I wish to express my appreciation to G. A. Gordon for his helpful criticisms and to Ann Marek for her editorial assistance throughout the manuscript preparation. I would like to thank H. C. Fritts for his useful suggestion on the use of a filter and also Linda Drew for her encouragement through the entire project.

The facilities of the University of Arizona Computer Center were used for the development of the programs.

REFERENCES

Baillie, M. G. L. and J. R. Pilcher
1973 A simple crossdating program for tree-ring research. *Tree-Ring Bulletin* 33: 7-14.

Douglass, A. E.
1935 Accuracy in dating — II. *Tree-Ring Bulletin* 1 (3) 19-21.

Duvick, D. N.
1978 The Border Beacon, Labrador, *Picea glauca* chronology. Technical Note No. 12. Laboratory of Tree-Ring Research, University of Arizona, Tucson, Arizona. 8 pp.

Ferguson, C. W.
1970 Concepts and techniques of dendrochronology. *In* "Scientific Methods in Medieval Archaeology" (R. Berger, ed.), pp. 183-200. University of California Press, Berkeley.

Fritts, H. C.
1963 Computer programs for tree-ring research. *Tree-Ring Bulletin* 25 (3-4) 2-7.
1976 *Tree rings and climate.* Academic Press, London. 567 pp.

Fritts, H. C., T. J. Blasing, B. P. Hayden, and J. E. Kutzbach
1971 Multivariate techniques for specifying tree-growth and climate relationships and for reconstructing anomalies in paleoclimate. *J. Appl. Meteorol.* 10 (5) 845-864.

Fritts, H. C., J. E. Mosimann, and C. P. Bottorff
1969 A revised computer program for standardizing tree-ring series. *Tree-Ring Bulletin* 29 (1-2) 15-20.

Giddings, J. L., Jr.
1942 Dated sites on the Kobuk River, Alaska. *Tree-Ring Bulletin* 9 (1) 1-8.

Julian, P. R. and H. C. Fritts
1968 On the possibility of quantitatively extending climatic records by means of dendroclimatological analysis. *Proc. First Statist. Meteorol. Conf.*, Amer. Meteorol. Soc., pp. 76-82, Hartford, Connecticut.

LaMarche, V. C., Jr. and H. C. Fritts
1972 Tree rings and sunspot numbers. *Tree-Ring Bulletin* 32: 19-33.

Lyon, C. J.
1953 Vertical uniformity in three New England conifers. *Tree-Ring Bulletin* 20 (2) 10-16.

Milsom, S. and M. Hughes
1977 Evaluation of x-ray densitometry as a technique. Paper presented at the International Symposium on Dendrochronology in Northern Europe, July 11-14, 1977, Greenwich, England.

Parker, M. L.
1967 Dendrochronology of Point of Pines. M. S. Thesis. University of Arizona, Tucson, Arizona. 168 pp.

Parker, M. L., G. M. Barton, and J. H. G. Smith
1976 Annual ring contrast enhancement without affecting x-ray densitometry studies. *Tree-Ring Bulletin* 36: 29-31.

Polge, H.
1966 Etablissement des courbes de variation de la densité du bois par expIloration densitométrique de radiographies d'échantillons prélevés à la tarière sur des arbres vivants. Application dans les domaines téchnologique et physiologique. *Ann. Sci. Forest* 23: 1-206.

Schulman, Edmund
1942 Variations between ring chronologies in and near the Colorado River drainage area. *Tree-Ring Bulletin* 8 (4) 26-32.
1944 Dendrochronology in Mexico, I. *Tree-Ring Bulletin* 10 (3) 18-24.

Stokes, M. A. and T. L. Smiley
1968 *An introduction to tree-ring dating.* University of Chicago Press, Chicago.

Wendland, W. M.
1975 An objective method to identify missing or false rings. *Tree-Ring Bulletin* 35: 41-47.

```
      SUBROUTINE SKEL1(OUT,ARRAY,ILAG,NRINGS,ICALL,CUTOFF)
COMMENTS ****************************************************************
C                            ***                                    C
C                   SUBROUTINE SKEL1                                C
C                   SEPTEMBER 1979                                  C
C                   JOHN PHILIP CROPPER                             C
C                   LABORATORY OF TREE RING RESEARCH                C
C                   UNIVERSITY OF ARIZONA                           C
C                   TUCSON,  ARIZONA. 85721    U.S.A.               C
C                            ***                                    C
COMMENTS ****************************************************************
C    PROGRAM TO TAKE THE RUNNING MEAN AND STANDARD DEVIATIONS OF AN
C    ARRAY ("ARRAY") AND DISTINGUISH SIGNIFICANTLY SMALLER OR/AND LARGER
C    RINGS. SMALL RINGS ARE SCALED BETWEEN ZERO AND TWO WHILE LARGE RINGS
C    ARE DESIGNATED "9.9".
C
C                   CALL CARD PARAMETERS                            C
C -------------------------------------------------------------------
C                            ***                                    C
C
C    OUT            OUTPUT
C        A MATRIX (WITH MAXIMUM OF 4 ROWS) THAT WILL CONTAIN THE SCALED
C        RESULTS.
C                            ***                                    C
C
C    ARRAY           INPUT
C        THE VECTOR CONTAINING THE DATA SET TO BE USED.
C                            ***                                    C
C
C    ILAG            INPUT
C        THE TOTAL NUMBER OF YEARS TO BE CONSIDERED IN THE CALCULATION
C        OF EACH MEAN.
C     ** MUST BE AN ODD NUMBER ** E.G. A 7 WOULD TAKE THE MEAN OF
C        SEVEN NUMBERS -- THAT IS 3 ON EITHER SIDE OF THE CENTRAL
C        VALUE BEING CONSIDERED.
C                            ***                                    C
C
C    NRINGS          INPUT
C        THE NUMBER OF VALUES INPUT IN THE VECTOR "ARRAY".
C        NOTE - THE COLUMNS OF THE MATRIX "OUT" MUST BE EQUAL TO OR
C        LARGER THAN NRINGS.
C                            ***                                    C
C
C    ICALL           INPUT
C        THE ROW OF THE OUTPUT MATRIX "OUT" IN WHICH YOU WISH THE
C        PRESENT RESULTS OF THE VECTOR "ARRAY" TO BE PLACED.
C        MUST BE AN INTEGER VALUE BETWEEN 1 AND 4
C        IT IS SUGGESTED THAT THE CALL TO THIS SUBROUTINE BE DONE IN A
C        "DO " LOOP AND THAT "ICALL" BE INCREASED BY ONE AT EACH CALL.
C                            ***                                    C
C
C    CUTOFF          INPUT
C        THE MULTIPLICATION FACTOR FOR THE NUMBER OF STANDARD
C        DEVIATIONS TO BE USED AS THE SIGNIFICANCE LEVEL. USUALLY A
C        REAL NUMBER BETWEEN 1. AND 0.5
C                            ***                                    C
C -------------------------------------------------------------------
C
C """"""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""
C                   VARIABLES USED                                  C
C                            ***                                    C
C
C    AMAX       - THE SMALLEST RING (OR THE LARGEST NEGATIVE DEVIATION FROM
C                 THE MEAN).
C    AMEAN      - VECTOR OF RUNNING MEANS.
C    IMISS      - THE NUMBER OF ELEMENTS OF "ARRAY" TO BE SKIPPED
C                 BEFORE STARTING THE CALCULATIONS.
C    ISTART     - THE FIRST ELEMENT OF "ARRAY" THAT CAN BE CONSIDERED.
C    ISTOP      - THE LAST ELEMENT OF "ARRAY" THAT CAN BE CONSIDERED.
C    KK AND KKK  THE COUNTERS THAT DELINEATE EACH GROUP OF YEARS TO BE
C                 ANALYSED AT A TIME.
C    STDEV      - VECTOR OF STANDARD DEVIATIONS OF THE ABOVE RUNNING MEANS.
C    SUM        - THE SUM OF "ILAG" ELEMENTS OF THE VECTOR "ARRAY".
C    SUMSQ      - THE SUMS OF SQUARES OF "ILAG" ELEMENTS OF VECTOR "ARRAY".
C    TEST       - THE VALUE USED TO TEST FOR SIGNIFICANTLY SMALLER RINGS.
C    TESTL      - THE VALUE USED TO TEST FOR SIGNIFICANTLY LARGE (BIG) RINGS.
C                            ***                                    C
```

Figure 4. FORTRAN program listing of subroutine SKEL1 that determines wide and narrow rings.

```
C ************************************************************************
C
      DIMENSION OUT(4,500), ARRAY(1), AMEAN(500), STDEV(500)
      DO 100 I = 1 , NRINGS
C
C     INITIALISE ARRAYS TO ZERO BEFORE STARTING COMPUTATION.
C
      AMEAN(I) = 0.
      STDEV(I) = 0.
  100 CONTINUE
C
C     CALCULATE START, STOP, AND INTERVAL PARAMETERS.
C
      AMAX = 0.
      ISTART = ( (ILAG - 1) / 2) + 1
      ISTOP = NRINGS - ( (ILAG - 1) / 2)
      IMISS = (ILAG - 1) / 2
         DO 200 I = ISTART,ISTOP
C
C        CALCULATE THE MEAN AND STDEV OF THE PERIOD AROUND THE RING.
C
      SUM = 0.
      SUMSQ = 0.
      KK = I - IMISS
      KKK= I + IMISS
              DO 300 J = KK,KKK
              SUM = SUM + ARRAY(J)
  300         SUMSQ = SUMSQ + (ARRAY(J)*ARRAY(J))
         AMEAN(I) = SUM / FLOAT(ILAG)
      STDEV(I) = SQRT((SUMSQ -(SUM*SUM)/FLOAT(ILAG))/FLOAT(ILAG))
  200    CONTINUE
      DO 400 I = ISTART,ISTOP
C
C     IDENTIFY THE RINGS THAT ARE MORE THAN ONE STDEV. SMALLER THAN THE MEAN
C     PUT ZERO"S IN "NORMAL" SIZED RING LOCATIONS OF THE ARRAY "OUT" AND
C     PUT THE DEPARTURE GREATER THAN THE STANDARD DEVIATION INTO THE
C     SMALL RING LOCATIONS OF THE "OUT" ARRAY.
C
      TEST = AMEAN(I) -(STDEV(I) * CUTOFF)
      TESTL = AMEAN(I) +(STDEV(I) * CUTOFF)
C     TEST FOR EXTRA LARGE RINGS AND PLACE AN IDENTIFING 9.9 IN THE
C     OUTPUT ARRAY "OUT".
      IF(ARRAY(I).LT.TESTL) GO TO 250
      OUT(ICALL,I) = 9.9
      GO TO 400
  250 IF(ARRAY(I).LE.TEST) GO TO 350
      OUT(ICALL,I) = 0.
      GO TO 400
  350   OUT(ICALL,I)   = TEST - ARRAY(I)
      IF(OUT(ICALL,I).GT.AMAX) AMAX = OUT(ICALL,I)
  400 CONTINUE
C
C     SET THE LAST NEW RINGS THAT CANNOT BE USED IN THE CALCULATIONS TO
C      ZERO BEFORE SCALEING THE REST OF THE SIGNIFICANTLY SMALLER RINGS.
C
             DO 450 I = ISTOP,NRINGS
  450        OUT(ICALL,I) = 0.0
      DO 500 I = 1,NRINGS
C     SCALE THE VECTOR IN THE RANGE ZERO TO POSITIVE TWO.
      OUT(ICALL,I) = OUT(ICALL,I) * (2.0 / AMAX )
  500 CONTINUE
      RETURN
      END
```

Figure 4, continued

```
      SUBROUTINE SKELET(OUT,NRINGS,FIRST,ISHRT,RID,HDG1)
C                            ***                                     C
COMMENTS *************************************************************
C                                                                    C
C                    SUBROUTINE SKELET                               C
C                    SEPTEMBER 1979                                  C
C                    JOHN PHILIP CROPPER                             C
C                    LABORATORY OF TREE RING RESEARCH                C
C                    TUCSON,  ARIZONA. 85721  U.S.A.                 C
C                                                                    C
COMMENTS *************************************************************
C                            ***                                     C
C     SUBROUTINE FOR PRODUCING SKELETON PLOTS ON A "CALCOMP" PLOTTER. THIS
C     USES THE DATA MATRIX OUTPUT FROM THE SUBROUTINE SKEL1.
C                            ***                                     C
C                    CALL CARD PARAMETERS
C                            ***                                     C
C     OUT                INPUT
C                A MATRIX (MAXIMUM OF 4 ROWS) OF SCALED RESULTS AS
C                PRODUCED BY "SKEL1".
C                            ***                                     C
C     NRINGS             INPUT
C                A VECTOR OF UP TO 4 ELEMENTS CONTAINING THE NUMBER OF DATA
C                VALUES IN EACH ROW OF MATRIX "OUT".
C                            ***                                     C
C     FIRST              INPUT
C                A VECTOR OF UP TO 4 ELEMENTS CONTAINING THE FIRST YEAR
C                ASSIGNED TO THE FIRST ELEMENT OF EACH ROW OF "OUT".
C                IT IS NOT USED IN THE SUBROUTINE AT THE MOMENT BUT CAN
C                BE USED WHEN PLOTTING ALREADY DATED MATERIAL.
C                            ***                                     C
C     ISHRT              INPUT
C                AN INTEGER VARIABLE SPECIFYING THE NUMBER OF ROWS IN "OUT"
C                THAT CONTAIN DATA. THIS CAN BE FROM 1 TO 4.
C                            ***                                     C
C     RID                INPUT
C                A VECTOR CONTAINING THE ID (6 CHARACTERS) OF EACH OF THE
C                DATA ROWS BEING PLOTTED.
C                            ***                                     C
C     HDG1               INPUT
C                A GENERAL HEADING (30 CHARACTERS) TO BE PLACED AT THE TOP
C                OF THE PLOT.
C                            ***                                     C
C                            ***                                     C
C                    VARIABLES USED.
C                            ***                                     C
C     AI         - THE DECADE TO BE WRITTEN (NUMBER NEEDS A "REAL"
C                  VARIABLE IN THE CALL STATEMENT)
C                            ***                                     C
C     HT         - IS THE HEIGHT OF THE CHARACTERS WRITTEN IN CALLS TO
C                  "SYMBOL" AND "NUMBER".
C                            ***                                     C
C     IBACK      - A BACK-SPACING COUNTER SO THAT THE DECADE LABELS ARE CENTERE
C                            ***                                     C
C     ITEST      - AN ALTERNATING COUNTER SO THAT NUMBERS ARE WRITTEN ONLY
C                  ON ALTERNATE DECADES MARKS.
C                            ***                                     C
C     MXRNG      - IS THE STORAGE SPACE FOR THE MAXIMUM LENGTH OF ANY GROUP
C                  FOUR PLOTS, SO THAT IT CAN BE USED FOR THE LENGTH OF THE
C                  BASE LINE.
C                            ***                                     C
C     NC AND NCM  COUNTERS FOR ANNUAL AND DECADE MARKS.
C                            ***                                     C
C     NSTOP      - THE LENGTH OF THE SERIES THAT IS BEING PLOTTED.
C                            ***                                     C
C     TML        - THE LENGTH OF THE DECADE TICK MARKS.
C                            ***                                     C
```

Figure 5. FORTRAN program listing of subroutine SKELET that uses CALCOMP plotting routines to produce skeleton plots (in groups of four).

```
C -----------------------------------------------------------------------
      DIMENSION OUT(4,500), NRINGS(1), FIRST(1), RID(1), HDG1(1)
C     CALL THE INITIALIZATION PLOTTING ROUTINE.
      CALL INITIAL(0,99,0.4,0)
C     IF "ISHRT" IS LESS THAN FOUR THEN THIS WILL ALTER THE OUTPUT FORMAT
      IF(ISHRT.GT.4) ISHRT = 4
      MXRNG = 0
      HT = 0.35
      TML  = -0.15
C     CONVERSION OF THE PLOTTING FROM INCHES TO CENTIMETERS.
      CALL FACTOR( 1.0/2.54)
C     SAVE PAPER BY MOVING BACK FROM THE COMPUTER SET ORIGIN, MOVE TO THE
C     TOP OF PAPER READY TO WRITE OUT THE HEADING -THIS IS THE NEW ORIGIN.
      CALL PLOT(-15.0,21.0,-3)
C     WRITE THE GENERAL HEADING THAT WAS ON THE CONTROL CARD, 30 CHARACTERS
      CALL SYMBOL(0.0,0.0,HT,HDG1,0.0,30)
C
C     DO THE PLOTS IN GROUPS OF FOUR.
      DO 145 I=1,ISHRT
C     WRITE THE CORE ID AT THE TOP LEFT HAND CORNER OF THE PLOT.
      CALL SYMBOL(0.0,-1.75,HT,RID(I),0.0,10)
C     "Y" IS THE SPACING FACTOR BETWEEN THE ID AND THE PLOT.
      Y = -4.0
C     SET "X" FOR THE BEGINNING OF A NEW PLOT.
C     I.E. GO TO THE LEFT HAND EDGE OF THE PAPER.
      X = 0.0
C
C     SET NEW ORIGIN  "Y" UNITS BELOW THE PRINTED ID.
      CALL PLOT(0.0,Y,-3)
C     WRITE THE FIRST TICK MARK.
C     DRAW A DECADE SIZED TICK MARK.
      CALL PLOT(0.0,TML,2)
C     TEST FOR THE LARGEST SERIES OF THIS BATCH BEING PLOTTED.
      IF(NRINGS(I).GT.MXRNG) MXRNG = NRINGS(I)
      NC = 0
      NCM = 10
      NSTOP = NRINGS(I)
C     PLOT THE ANNUAL AND DECADE TICK MARKS IN THE Y AXIS ONLY.
          DO 140 J=1,NSTOP
C     IF  J.EQ.1  PLOT THE FIRST VALUE AS THE TICK MARK IS ALREADY DRAWN.
          IF(J.GT.1) GO TO 130
          GO TO 131
  130     NC = NC + 1
C     INCREMENT YEAR COUNTER AND DISTANCE MOVED ALONG "X" AXIS.
          X = X + 0.2
C     TEST FOR A DECADE OR THE LAST RING OF THE SERIES.
          IF(NC.EQ.NCM.OR.J.EQ.NRINGS(I))  GO TO 132
C
C     MOVES THE PEN OFF THE PAPER TO THE NEXT POSITION ON THE X AXIS.
C     PUT A SMALL MARK FOR EACH ANNUAL VALUE.
C     MOVE PEN TO BELOW "X" AXIS WITH PEN OFF THE PAPER.
  131 CALL PLOT(X,-0.05,3)
C     DRAW THE ANNUAL MARK.
      CALL PLOT(X,0.0,2)
C     LIFT PEN OFF PAPER.
          CALL PLOT(X,0.0,3)
          GO TO  135
C     INCREMENT DECADE COUNTER.
  132     NCM = NCM + 10
C     DRAW IN THE TICK FOR EACH DECADE AS YOU GET TO IT.
C     MOVE PEN BELOW "X" AXIS READY FOR PLOTTING DECADE MARK.
          CALL PLOT(X,TML,3)
C     PLOT MARK.
      CALL PLOT(X,0.0,2)
C     LIFT PEN OFF THE PAPER.
      CALL PLOT(X,0.0,3)
C     TEST FOR BIG RINGS.
  135 IF(OUT(I,J).LT.9.) GO TO 136
C     PLOT A LETTER "L" FOR BIG RINGS.
      CALL SYMBOL(X,2.0,0.15,1HL,0.0,1)
      GO TO 140
```

Figure 5, continued

```
C     PLOT THE DATA AT EACH POINT AS YOU GET TO IT.
  136     CALL PLOT(X,OUT(I,J),2)
  140     CONTINUE
C     DRAW IN THE X-AXIS ON THE WAY BACK TO THE ORIGIN.
      CALL PLOT(X,0.0,3)
      CALL PLOT(0.0,0.0,2)
  145 CONTINUE
C
C     SET "X" FOR THE BEGINNING OF A NEW PLOT.
      X = 0.0
C     PUT IN A BASE LINE WITH ALTERNATE DECADES LABELED.
C     FORM A NEW ORIGIN.
      CALL PLOT(0.0,-1.5,-3)
      DO 150 I=1,MXRNG,10
C     PUT IN THE DECADE TICKS.
      CALL PLOT(X,TML,3)
      CALL PLOT(X,0.0,2)
C     INCREMENT "X" TO MOVE IT TEN YEAR DISTANCE AT A TIME.
  150 X = X + 2.0
C     DRAW IN THE BASE LINE ON THE WAY BACK TO THE ORIGIN.
          CALL PLOT(0.0,0.0,2)
C     FORM A NEW ORIGIN FOR THE PLOTTING OF THE NUMBERS.
      CALL PLOT(0.0,-.75,-3)
      ITEST = 0
      DO 160 I = 1,MXRNG,10
C     TO CENTRE UP THE DACADE NUMBERS THE FOLLOWING IF STATEMENTS ARE USED.
      IF(I.LE.9) IBACK = 2
      IF(I.GE.10.AND.I.LE.99)  IBACK = 2
      IF(I.GE.100.AND.I.LE.999)  IBACK = 3
      IF(I.GE.1000)  IBACK = 4
      II = I -IBACK
      X = 0.2 * FLOAT(II)
      AI = FLOAT(I-1)
C     PLACE THE NUMBERS BELOW THE BASE LINE.
C     TEST SO THAT ONLY ALTERNATE DECADES ARE LABELED.
      IF(ITEST.EQ.0) CALL NUMBER(X,0.0,HT,AI,0.0,-1)
      IF(ITEST.EQ.0) GO TO 155
      ITEST = 0
      GO TO 160
  155 ITEST = 1
  160 CONTINUE

C     TERMINATE THIS CALL TO THE PLOTTER.
      CALL ENDPLT
      RETURN
      END
```

Figure 5, continued